Principles of Computable General Equilibrium (CGE) Modeling for Climate Policy Analysis

Niven Winchester

October 4, 2013

EPPA Workshop
Sunday River, Newry, ME

Outline

1. An introduction to general equilibrium modeling
2. An introduction to computable general equilibrium modeling
3. A CGE model solved in GAMS as a mixed complementarity problem
4. An introduction to the Mathematical Programming System for General Equilibrium Analysis
5. An illustrative energy-economic computable general equilibrium model
6. Adding an advanced technology to a computable general model
7. A global model for climate policy analysis

Outline

Topic	Coverage	Model(s)
1. An introduction to general equilibrium modeling	Full	
2. An introduction to computable general equilibrium modeling	Partial - solutions to exercises provided	
3. A CGE model solved in GAMS as a mixed complementarity problem	Partial	SIMPLE0.gms
4. An introduction to the Mathematical Programming System for General Equilibrium Analysis	Partial - supplementary notes provided	SIMPLE1.gms, SIMPLE2.gms, SIMPLE3.gms
5. An illustrative energy-economic computable general equilibrium model	Partial - supplementary notes provided	ENERGY1.gms, ENRGY2.gms
6. Adding an advanced technology to a computable general model	Full	ENERGY3.gms
7. A global model for climate policy analysis	Partial	GLOBAL.gms

1. An introduction to general equilibrium modeling

What is a general equilibrium model?

- An individual market is said to be in equilibrium when the quantity of a good demanded is equal to the quantity supplied
- When all markets are in equilibrium, the economy is said to be in a state of general equilibrium
- General equilibrium deals explicitly with the interrelationships between different sectors of the economy
- General equilibrium models consider equilibrium in all markets simultaneously – consider interrelationships between different sectors in the economy – and explicitly represent the circular flow of income

What is a general equilibrium model?

A general equilibrium model includes:

1. Firms that maximize profits/minimize costs
2. Households who maximize welfare/utility by demanding commodities according to prices
3. Markets that mediate behavior of economic agents (e.g., prices adjust until supply and demand are equal)
4. A government that collects taxes and spends revenue on consumption and transfers to households

What is a general equilibrium model?



MIT JOINT PROGRAM
GLOBAL CHANGE

What is a general equilibrium model?

- The vision of an economy as a vast system of interlocking markets was first put forward by Leon Walras (1834-1910)
- In general equilibrium, the supply and demand functions in both product and factor markets will typically be related to all the prices in the system, and the income and expenditure relations of all the agents in the economy – firms, households and government – have to be explicitly modeled
- If there are N commodity markets and F factor prices, there are $N + F$ markets and $N + F$ equilibrium prices to be determined

The structure of a simple general equilibrium model

- *N* commodities: Y_1,..., Y_N with prices $p_1, \dots, p_N$
- *F* factors: $K_1{}^*$,...,$K_F{}^*$ with prices $r_1, \dots, r_F$
- Perfect competition⇨ each industry can be represented by a single producer employing a constant returns to scale production function: $Y_i = Y_i(\gamma K_{1i}, \dots, \gamma K_{Fi})$ =$\gamma Y_i(K_{1i}, \dots, K_{Fi})$, where K_{fi} *is* demand for factor *f* by firm *i*
- A representative household/consumer earns income (*M*) from supplying production factors to firms and demands commodities C_1,..., C_N to maximize utility: $U = U(C_1, \dots, C_N)$
- Static model (no time dimension) of a closed economy with no government
- Factors are endowed (i.e., not produced)

GLOBAL CHANGE

The structure of a simple general model



Competitive equilibrium

- A competitive equilibrium consists of a non-negative consumption plan $\{C_1 \geq 0,\ldots, C_N \geq 0\}$ for the representative consumer, a non-negative production plan for each firm $\{Y_1 \geq 0,\ldots, Y_N \geq 0; K_{11} \geq 0,\ldots, K_{FN} \geq 0\}$ and non-negative prices $\{p_1 \geq 0, \ldots, p_N \geq 0; r_1 \geq 0, \ldots, r_F \geq 0\}$ such that:

1. **Utility maximization:** The consumption bundle maximizes the utility of the representative consumer given the budget constraint
2. **Profit maximization:** For each firm, the set of inputs used maximizes profits given the firm's production technology
3. **Market clearing:** For each market, demand does not exceed supply

Equilibrium as a constrained maximization problem

- Equilibrium in this economy is the solution to the following constrained optimization problem:

$$\text{Max } U = U(C_1, \ldots, C_N)$$

subject to:

$$Y_i \leq Y_i(K_{1i}, \ldots, K_{Fi})$$

$$K_f{}^* \geq \sum_{i=1}^{N} K_{fi} \quad \forall\, f$$

$$Y_i = C_i \quad \forall\, i$$

Equilibrium as a constrained maximization problem

- While solving for the equilibrium as a constrained optimization problem works for simple problems, the usefulness of this approach breaks down quickly as the model becomes more complicated:
  - Multiple households and/or multiple countries ⇨ what is the objective function?
  - Incorporation of "second-best" features is not straightforward
  - Tedious algebra for large-scale models
- **An alternative approach is to define the problem as the solution to a system of equations that embodies the underlying optimization behavior**
  - We will follow this approach in the remainder of the workshop

Equations in a general equilibrium model

Consumer behavior

Maximize $U = U(C_1, \ldots, C_N)$

subject to $\sum_{i=1}^{N} p_i C_i = M$

$\Rightarrow C_i = C_i(p_1, \ldots, p_N, M)$ (*N* equations)

where C_i is demand for product *i* (*i* = 1,…, *N*), p_i is the price of product *i*, and *M* is consumer income.

Equations in a general equilibrium model

Firm behavior

Minimize $\sum_{f=1}^{F} r_f K_{fi}$

subject to $Y_i = Y_i(K_1, \dots, K_F)$

$\Rightarrow K_{fi} = K_{fi}(r_1, \dots, p_F, Y)$ (*F*×*N* equations)

and $k_{fi} = \frac{K_{fi}}{Y_i} = k_{fi}(r_1, \dots, p_F)$ (*F*×*N* equations)

where Y_i is the output of firm i, K_{fi} *is* demand for factor *f* (*f* = 1,…, *F*) by firm *i*, r_f is the price of factor *F*, and k_{fi} is the unit factor demand function for factor *f* by firm *i*.

Equations in a general equilibrium model

Zero profit conditions

$$p_i = \sum_{f=1}^{F} r_f k_{fi}$$ (*N* equations)

Consumer income

$$M = \sum_{f=1}^{F} r_f K_f^{\,*}$$ (1 equation)

Product market clearing

$$Y_i = C_i$$ (*N* equations)

Factor market clearing

$$K_f^{\,*} = \sum_{i=1}^{N} K_{fi}$$ (*F* equations)

where $K_f^{\,*}$ is the endowment of factor *f*.

Equations in a general equilibrium model

COMMODITY MARKETS		
Demand	$C_i = C_i(p_1, \dots, p_N, M)$	(N equations)
Zero profit	$p_i = \sum_{f=1}^{F} r_f k_{fi}$	(N equations)
Market clearing	$Y_i = C_i$	(N equations)
FACTOR MARKETS		
Demand	$K_{fi} = K_{fi}(r_1, \dots, p_F, Y)$	(F×N equations)
	$k_{fi} = \dfrac{K_{fi}}{Y_i} = k_{fi}(r_1, \dots, p_F)$	(F×N equations)
Market clearing	$K_f^{\,*} = \sum_{i=1}^{N} K_{fi}$	(F equations)
CONSUMER INCOME		
	$M = \sum_{f=1}^{F} r_f K_f^{\,*}$	(1 equation)
Endogenous variables	$C_1 \dots C_N;\ Y_1 \dots Y_N;\ K_1 \dots K_F;\ k_1 \dots k_F;$ $p_1 \dots p_N;\ r_1 \dots r_F;\ M$	($3N$ + $2FN$ + F + 1 variables)
Exogenous variables	$K_1^{\,*}, \dots, K_F^{\,*}$	(F variables)

Walras' law and the need for a *numeraire*

- The model is a system of equations with $3N + 2FN + F + 1$ ($= q$) endogenous variables and $3N + 2FN + F + 1$ equations
- Walras' law states that for a given set of prices, the sum of the excess demand over all markets must be equal to zero (see lecture example – A proof of Walras' law)
- Implies that if $N + F - 1$ markets are in equilibrium, the $(N + F)^{th}$ market will automatically be in equilibrium
- **Problem:** We only have q-1 independent equations (and q endogenous variables)

Walras' law and the need for a *numeraire*

- **Solution:** Only relatives prices matter. If all prices (including factor prices) change by the same proportion, the solution does not change
  - There is no money illusion
  - Demand functions are homogenous of degree zero with respect to prices
- It is convenient to set the price of one good or factor equal to one and express all other prices relative to that price
- The good with the price set equal to one is known as the *numeraire* commodity
- When a *numeraire* is chosen, the model is a system of q-1 independent equations and q-1 endogenous variables

Lecture examples

1. The equations of a Cobb-Douglas 2×2 general equilibrium model
   - Derive the system of equations for a general equilibrium model with two factors, two goods and Cobb-Douglas production and utility functions
2. A proof of Walras' law
   - Show that Walras' law holds for *any* combination of prices
3. A pure exchange general equilibrium model
   - Solve a general equilibrium model (by hand) with two goods, two consumers and endowed commodities

A general equilibrium model with intra-industry flows

- In addition to factors of production, industries use inputs from other industries to produce output (e.g., the electricity industry uses inputs of coal and gas)



A general equilibrium model with intra-industry flows

- When intermediate inputs are considered, the production functions are given by:

$$Y_i = Y_i(Y_1, \dots, Y_{N;} K_1, \dots, K_F)$$

- If intermediate inputs are used in fixed proportion to output, the zero profit conditions are:

$$p_i = \sum_{j=1}^{N} a_{ij} p_j + \sum_{f=1}^{F} r_f k_{fi}$$

where a_{ij} is the input requirement of *j* per unit of good *i*

GLOBAL CHANGE

A general equilibrium model with intra-industry flows

- We also need to change the market clearing equations to reflect demand for intermediate inputs:

$$Y_i = C_i + \sum_{j=1}^{N} a_{ji} Y_j$$

- **Note:** Adding intermediate inputs does not change the number of equations or the number of endogenous variables

2. An introduction to computable general equilibrium modeling

What is a computable general equilibrium model?

- Only very simple general equilibrium models can be solved analytically
- Computable general equilibrium (CGE) models provide a numerical solution to the equations of a general equilibrium model by solving for the set of prices that achieves equilibrium in all markets
- CGE models facilitate counterfactual quantitative policy analysis (i.e., how might an economy react to changes in policy, technology, or other external factors)
  - Useful for evaluating policy changes applied to many sectors, or when a change in one sector has a large impact on other sectors of the economy
- CGE models are calibrated and benchmarked to observed socioeconomic data so that they are able to replicate the observed (benchmark) data endogenously

A social accounting matrix (SAM)

- Data for a CGE model are organized using a social accounting matrix (SAM)
- A SAM represents data on all economic transactions that take place within an economy
- Numbers in a SAM represent the values (price times quantity) of economic transactions at a point in time (e.g., within a year)
- A SAM provides a matrix representation of the national accounts
- When calibrating a CGE model, all prices are normalized to one
- Choose quantity units so that all prices equal one (e.g., gasoline is measured in ¼ gallon units)

A social accounting matrix (SAM)

- A SAM represents an equilibrium where zero profit, market clearing and income balance conditions hold
- In a SAM, columns correspond to the production sectors in the economy and the representative consumer/household, and rows correspond to product and factor markets
- A column sum equal to zero implies that profit is zero and a row sum implies that the market clears
- If the SAM represents an equilibrium, all column sums and row sums should equal zero
- In each column, positive values represent income and negative values represent expenditure

A 2×2 Social Accounting Matrix

	Y_1	Y_2	HH	Row sum
p_1	100		-100	0
p_2		100	-100	0
w	-25	-75	100	0
r	-75	-25	100	0
Col. sum	0	0	0	

where, for goods 1 and 2, Y_1 and Y_2 are output quantities, and p_1 and p_2 are commodity prices; and w and r are, respectively, the price of labor and capital

Functional forms

- Constant elasticity of substitution (CES) functions are a convenient way of characterizing production and consumption tradeoffs

$$Y = \gamma\left(\sum_{f=1}^{F} \alpha_f K_f^{\frac{(\sigma-1)}{\sigma}}\right)^{\frac{\sigma}{(\sigma-1)}}$$

where σ is the elasticity of substitution

- Special cases for CES function
  - $\sigma = 0$: Perfect complements
  - $\sigma = 1$: Cobb-Douglas
  - $\sigma = \infty$: Perfect substitutes

Functional forms

- Nesting CES functions allows substitution possibilities to differ between different inputs/commodities

$$Y \quad = f\{\sigma_A; K_1, g(\sigma_B; K_2, K_3)\}$$



Calibration

- Calibration involves assigning production and utility function parameters so that the model replicates the benchmark data

- For demand functions derived from Cobb-Douglas utility functions ($C_i = \frac{\alpha_i M}{p_i}$) it is easy to show that:

  $\alpha_i = \frac{\bar{p}_i \bar{C}_i}{M}$ where bars denote benchmark prices and quantities

Calibration

- For demand functions derived from CES utility and production functions, parameters can also be obtained by inverting the demand functions (and substitution) but doing so is tedious and error-prone

- It is more convenient to work with the (equivalent) calibrated share form of the CES function

$$Y \;\; = \bar{Y}\left(\sum_{f=1}^{F} \theta_f {K_f}^{\frac{(\sigma-1)}{\sigma}}\right)^{\frac{\sigma}{(\sigma-1)}}$$

where $\theta_f = \dfrac{\bar{r}_f \bar{K}_f}{\sum_{f=1}^{F} \bar{r}_f \bar{K}_f}$ (i.e., the cost share of factor *f*)

Calibration

- The associated cost and demand functions for the CES production function in calibrated share form are:

$$C(\boldsymbol{r}) = \bar{C}\left(\sum_{f=1}^{F} \theta_f \left(\frac{r_f}{\bar{r}_f}\right)^{(1-\sigma)}\right)^{\frac{1}{(1-\sigma)}}$$

$$K_f = \bar{r}_f \frac{Y}{\bar{Y}}\left(\frac{C\bar{r}_f}{\bar{C}r_f}\right)^{\sigma}$$

Calibration

- A CGE model with CES production and utility functions in calibrated share form can be calibrated from:
  - Benchmark factor demands (values)
  - Benchmark factor prices (usually set equal to one)
  - Benchmark costs (values)
  - Benchmark output (values)
  - Benchmark value shares
  - **Assigned elasticities of substitution**

Equilibrium as a mixed complementarity problem

- Rutherford (1995) and Mathiesen (1985) show that many economic models, including general equilibrium models, can be cast as a mixed complementary problem
- Mixed complementarity problem (MCP)
  - Given a function $F(z)$, find z such that $F(z) \geq 0$, $z \geq 0$ and $zF(z) = 0$
  - Example: $F(z) = 5 - z \Rightarrow zF(z) = \mathrm{z}(5 - \mathrm{z})$

    $z = 0 \Rightarrow F(z) \neq 0$

    $z = 5 \Rightarrow F(z) = 0$
- "Mixed": solution is a mix of equalities, $F(z) = 0$, and inequalities, $z \geq 0$.
- "Complementarity" z and $f(z)$ are a complementary pair. z is an associated variable to a certain condition.

MIT JOINT PROGRAM
SCIENCE POLICY
GLOBAL CHANGE

Equilibrium conditions in MCP format

- An equilibrium in complementarity format is represented by a non-negative vector of activity levels, a non-negative vector of prices, and a non-negative vector of incomes such that:

1. Zero profit conditions: No production activity makes a negative profit, output is an associated variable

   – profit $\geq$ 0, output $\geq$ 0, output(–profit) = 0

2. Market clearance conditions: Excess supply (supply minus demand) is non-negative for all goods and factors, price is an associated variable

   supply – demand $\geq$ 0, price $\geq$ 0, price(supply – demand) = 0

3. Income definition: Expenditure equals income

   income = value of endowments

Equilibrium conditions in MCP format

- Household utility can be expressed as production activity, where consumption goods are assembled to produce utility/welfare (W) at minimum expenditure
  - Can be modeled in the same way as other produced commodities, where the expenditure function is analogous to the cost function
- Notation:
  - $c_i(r_1, \ldots, r_F)$ is the unit cost function
  - $k_{if}(r_1, \ldots, r_F)$ are the unit factor demand functions
  - $e(p_1, \ldots, p_n)$ is the unit expenditure function
  - $d_i(p_1, \ldots, p_n)$ are the unit compensated/Hicksian demand functions

Equilibrium conditions in MCP format

Equation	Associate variable
Non-positive profits for commodities:	
$-(p_i - c_i(r_1, \dots, r_F)) \geq 0$	Y_i
Non-positive profits for W:	
$-(p_w - e(p_1, \dots, p_N)) \geq 0$	W
Market clearing for commodities:	
$Y_i - d_i(p_1, \dots, p_N)W \geq 0$	p_i
Market clearing for W:	
$W - \dfrac{M}{p_w} \geq 0$	p_w
Factor market clearing:	
$K_f - \sum_{i=1}^{N} k_{fi} Y_{fi} \geq 0$	r_f
Income balance:	
$M = \sum_{f=1}^{F} r_f K_f$	

GLOBAL CHANGE

Equilibrium conditions in MCP format

- Zero profit conditions
  - If a zero profit condition holds as a strict inequality in equilibrium (i.e. profits for that activity are negative) that activity is not used
  - If a zero profit conditions holds with equality, its associated activity level is positive
- Market clearing conditions
  - If a market-clearing condition holds as a strict inequality, supply exceeds demand for that good or factor in equilibrium so its price must be zero
  - If a market-clearing condition holds with equality, the price of the associated good is strictly positive
- There are “off-the-shelf” algorithms for solving complementarity problems (e.g. GAMS/MILES and GAMS/PATH solver)

Conducting a CGE analysis



3. A CGE model solved in GAMS as a mixed complementarity problem

A 2×2 model in MCP format

- The file `SIMPLE0.gms` contains General Algebraic modeling System (GAMS) code for a 2x2 CGE model
- Production and utility functions are represented by CES functions, where the elasticity of substitution is 0.5 in all functions
- The model is based on the following SAM

	X	Y	W	$CONS$	$rowsum$
p_X	100		-100		0
p_Y		100	-100		0
p_W			200	-200	0
p_L	-25	-75		100	0
p_K	-75	-25		100	0
$colsum$	0	0	0	0	0

A 2×2 model in MCP format

```
* Declare elasticity parameters:

PARAMETERS
sigma_x    "Elasticity of substitution between capital and labor (X sector)" /
sigma_y    "Elasticity of substitution between capital and labor (Y sector)" /
sigma_w    "Elasticity of substitution between X and Y (W sector)" /.5/
lendow     "Labor endowment multiplier" /1/;

* Declare variables and equations for the 2x2 model:

POSITIVE VARIABLES
        X       X sector output index
        Y       Y sector output index
        W       Welfare index
        PX      Price index for commodity X
        PY      Price index for commodity Y
        PL      Price index for primary factor L
        PK      Price index for primary factor K
        PW      Price index for welfare
        HH      Household income and expenditure;
```

A 2×2 model in MCP format

```
EQUATIONS
        PRF_X   Zero profit for sector X
        PRF_Y   Zero profit for sector Y
        PRF_W   Zero profit for sector W

        MKT_X   Supply-demand balance for commodity X
        MKT_Y   Supply-demand balance for commodity Y
        MKT_L   Supply-demand balance for primary factor L
        MKT_K   Supply-demand balance for primary factor L
        MKT_W   Supply-demand balance for aggregate demand

        I_HH  Income definition for HH;
```

A 2×2 model in MCP format

```
*   Zero profit conditions:
PRF_X..        (25/100)*PL**(1-sigma_x) + (75/100)*PK**(1-sigma_x)
                                                   =G= PX**(1-sigma_x);
PRF_Y..        (75/100)*PL**(1-sigma_y) + (25/100)*PK**(1-sigma_y)
                                                   =G= PY**(1-sigma_y);
PRF_W..        (100/200)*PX**(1-sigma_w) + (100/200)*PY**(1-sigma_w)
                                                   =G= PW**(1-sigma_w);

*   Market clearing conditions:
MKT_X..        100 * X =E=  0.5*(PW/PX)**sigma_w * W*200;
MKT_Y..        100 * Y =E=  0.5*(PW/PY)**sigma_w * W*200;
MKT_W..            200 * W =E= HH / PW;
MKT_L..        100 * lendow =G= 0.25*100*X*(PX/PL)**sigma_x
                                  + 0.75*100*Y*(PY/PL)**sigma_y;
MKT_K..        100 =G= 0.75*100*X*(PX/PK)**sigma_x
                         + 0.25*100*Y*(PY/PK)**sigma_y;

*   Income definition:
I_HH..         HH =E= 100*lendow*PL + 100*PK;
```

GLOBAL CHANGE

A 2×2 model in MCP format

```
*   Define model equations and assign variables to equations:
MODEL SIMPLE_MCP /PRF_X.X, PRF_Y.Y, PRF_W.W, MKT_X.PX, MKT_Y.PY, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW, I_HH.HH /;

*   Initialize variables:

X.L=1;Y.L=1;W.L=1;PW.l=1;PX.L=1;PY.L=1;PK.L=1;PL.L=1;PW.L=1;HH.L=200;

*   Set lower bounds for variables to prevent execution errors:

PY.LO=1E-10;PW.LO=1E-10;PX.LO=1E-10;PK.LO=1E-10;PL.LO=1E-10;

*   Solve statement:

SIMPLE_MCP.ITERLIM = 0;
SOLVE SIMPLE_MCP USING MCP;
```

A 2×2 model in MCP format

	LOWER	LEVEL	UPPER	MARGINAl
---- VAR X	.	1.0000	+INF	.
---- VAR Y	.	1.0000	+INF	.
---- VAR W	.	1.0000	+INF	.
---- VAR PX	1.000000E-10	1.0000	+INF	.
---- VAR PY	1.000000E-10	1.0000	+INF	.
---- VAR PL	1.000000E-10	1.0000	+INF	.
---- VAR PK	1.000000E-10	1.0000	+INF	.
---- VAR PW	1.000000E-10	1.0000	+INF	.
---- VAR HH	.	200.0000	+INF	.

Exercises

1. Check the benchmark calibration. Introduce an imbalance in (a) one of the market clearance equations, in (ii) one of the zero-profit equations and diagnose the error using the `.lst` file.

2. Increase the labor endowment of the economy by 20% relative to the benchmark. Explain the observed outcomes.

3. Increase the labor endowment of the economy by 20% relative to the benchmark, and vary the elasticity of substitution. Explain the observed outcomes.

4. An introduction to the Mathematical Programming System for General Equilibrium (MPSGE) Analysis

Supplementary course notes:
An introduction to the Mathematical Programming System for General Equilibrium Analysis.pdf

What is MPSGE?

- The Mathematical Programming System for General Equilibrium Analysis (MPSGE) operates as a subsystem to the mathematical programming language in GAMS
- MPSGE simplifies specification of CGE models by freeing modelers from tedious equation writing

The structure of MPSGE models

- An MPSGE model is specified within an `$ONTEXT/$OFFTEXT` block
- The first keyword following `$ONTEXT` must define the model name

```
$ONTEXT
$MODEL:SIMPLE1
```

- There are four classes of variables in an MPSGE model

Variable	Description
`$SECTORS`	Production sectors (output indexes)
`$COMMODITIES`	Prices of produced commodities and endowed factors
`$COMSUMERS`	Consumers
`$AUXILLIRAY`	Endogenous taxes and endowment multipliers (optional)

A simple CGE model (SIMPLE1.gms)

```
$MODEL:SIMPLE1

$SECTORS:
        X               ! X sector output index
        Y               ! Y sector output index
        W               ! Welfare index

$COMMODITIES:
        PX              ! Price index for commodity X
        PY              ! Price index for commodity Y
        PL              ! Price index for primary factor L
        PK              ! Price index for primary factor K
        PW              ! Price index for welfare

$CONSUMERS:
        HH              ! Household income and expenditure
```

A simple CGE model (`SIMPLE1.gms`)

- A production block must be defined for each activity level
  - Each production block represents a constant elasticity of substitution (CES) function
  - The first line of a production block begins with the `$PROD` command, followed by a sector name, and assignment of the elasticity of substitution (following `s:`)
  - "`O:`" denotes an output field and "`I:`" denotes an input field
  - Price variables are used to indicate which commodity is being produced, and which commodities are used as inputs
  - Quantities follow a "`Q:`" statement

```
$PROD:X s:1
    O:PX        Q:100
    I:PL        Q:25
    I:PK        Q:75
```

```
$PROD:Y s:1
    O:PY        Q:100
    I:PL        Q:75
    I:PK        Q:25
```

```
$PROD:W s:1
    O:PW        Q:200
    I:PX        Q:100
    I:PY        Q:100
```

A simple CGE model (SIMPLE1.gms)

- A $DEMAND block must be included for each consumer
  - Commodities demanded follow a "D:" statement, and income from endowment commodities follow an "E:" statement

```
$DEMAND:HH
        D:PW          Q:100
        E:PL          Q:100
        E:PK          Q:100
```

- Following the $OFFTEXT statement denoting the end of the MPSGE syntax, GAMS is instructed to compile the functions and set a numeraire (if a numamaire is not set, a default numeraire will be chosen)

```
$SYSINCLUDE mpsgeset SIMPLE1

*       Set a numeraire
PL.FX = 1;
```

A simple CGE model (SIMPLE1.gms)

- Solving the model without introducing any shocks allows the user to check the consistency of the benchmark dataset (i.e., that the social accounting matrix underpinning the model is balanced)
- If there is an inconsistency in the benchmark dataset, setting the iteration limit equal to zero can be used to determine the nature of the inconsistency

```
SIMPLE1.iterlim = 0;
$INCLUDE SIMPLE1.GEN
SOLVE SIMPLE1 USING MCP;
```

- When the iteration limit is zero: The benchmark dataset is consistent if the MARGINAL value for all variables is zero (“.”)
  - A non-zero MARGINAL value for an activity level indicates non-zero profits
  - A non-zero MARGINAL value for a price variable indicates that the market is not in equilibrium

A simple CGE model (SIMPLE1.gms)

- Consistency of the benchmark data can also be checked by solving the model when the iteration limit is positive

```
SIMPLE2.iterlim = 1000;
$INCLUDE SIMPLE2.GEN
SOLVE SIMPLE2 USING MCP;
```

- When the iteration limit is positive: The benchmark dataset is consistent if the LEVEL of all activity and price variables is 1

Debugging MPSGE models

- If there is a GAMS programming error, GAMS will return the message:

```
****Status: Compilation errors
```

- If there is a MPSGE programming error, GAMS will return the message:

```
****MPSGE program error
```

- The source of the error can be determined by searching for "****" in the listing file.

Exercises (`SIMPLE1.gms`)

1. Change the quantity of `X` output from 100 to 110 and solve the model. Interpret the listing file.
2. Change `PL` in the `$DEMAND` block to `PK` and solve the model. Interpret the error message in the listing file.

- Essential topics not covered (see supplementary notes)
  - Set notation
  - Reporting variables
  - Representing exogenous taxes

An endogenous tax (`SIMPLE3.gms`)

- The `$AUXILIARY` command can be used to define endogenous variables that control certain outputs in the model
- We will use an endogenous tax on capital in the X sector to target a predetermined level of X output
- Define GAMS parameters to control the endogenous tax

```
PARAMETER
tflag(f,i)        Endogenous tax flag
ytarg(i)          Target output level;
```

- Define the tax as an auxiliary variable in the MPSGE code

```
$AUXILIARY:
TAU(f,i)$tflag(f,i)        ! Endogenous factor tax
```

- Include the endogenous tax in the MPSGE production block

```
$PROD:Y(i) s:1
  O:PY(i)          Q:vom(i)
  I:PF(f)          Q:vfm(f,i)  A:HH   N:TAU(f,i)$tflag(f,i)
```

An endogenous tax (`SIMPLE3.gms`)

- Include a `$CONSTRAINT` command in the MPSGE code

```
$CONSTRAINT:TAU(f,i)$tflag(f,i)
Y(i) =e= ytarg(i);
```

- Solving the model for an endogenous tax by: (i) turning on the `tflag(f,i)`, setting a target output level, and assigning an initial level of the tax

```
tflag("cap","axes") = YES;
ytarg("axes") = 0.95;
tau.L("cap","axes") = 0.2;
```

- The default lower limit for `$AUXILIARY` variables is zero and the default upper limit is infinity. Alternative upper and lower limits can be assigned using the "`.lo`" and "`.up`" suffixes (e.g, `tau.up("cap","axes") = 1`)

- **Exercise:** Implement an endogenous tax on capital in the X sector that increases X output by 10%.

5. An illustrative energy-economic computable general equilibrium model

Supplementary course notes:
An illustrative energy-economic computable general equilibrium model

A simple energy-economic model (ENERGY2.gms)

- `ENERGY2.gms` identifies five energy sectors (coal, crude oil, refined oil, gas and electricity), groups other sectors in a single sector, and considers CO_2 emissions from the combustion of fossil fuels

```
SET i Sectors /
   cru          Crude oil
   oil          Refined oil
   col          Coal
   gas          Gas
   ele          Electricity
   oth          Other output/;

SET f Factors of production /
   lab          Labor
   cap          Capital
   res          Natural resources /;

ALIAS (i,j);

SET ele(i) Electricity             /ele/;
SET ff(i)  Fossil fuels            /cru,col,gas/;
SET fe(i)  Final energy            /oil,col,gas/;
SET en(i)  Energy markets          /gas,ele,oil,col,cru/;
```

A simple energy-economic model (`ENERGY2.gms`)

- Data for the model is sourced from `edata.gms`

```
$include edata.gms
```

Input/output	Sector						
	Gas	Electricity	Oil	Coal	Crude Oil	OTHR	HH
Gas	150	-60				-70	-20
Electricity		850				-750	-100
Oil		-30	500			-460	-10
Coal		-35		50		-10	-5
Crude Oil			-300		300		
OTHR	-65	-325	-175	-15	-170	24790	-24040
Capital	-50	-300	-15	-10	-70	-9500	9945
Labor	-20	-100	-10	-10	-10	-14000	14150
Gas resource	-15						15
Coal resource				-15			15
Crude oil resource					-50		50

A simple energy-economic model (ENERGY2.gms)

```
*       Fossil fuel production (crude oil, coal, gas)
$PROD:Y(ff)$vom(ff) s:sigma("res_oth") b:0 va(b):sigma("k_l")
        O:PY(ff)        Q:vom(ff)
        I:PY(j)         Q:vfi(j,ff)       b: a:HH T:tf(j,ff)
        I:PL            Q:vfm("lab",ff)  va:
        I:PK            Q:vfm("cap",ff)  va:
        I:PR(ff)        Q:vfm("res",ff)
```



Fossil fuels

A simple energy-economic model (ENERGY2.gms)

```
*       Electricity production - an equivalent specification
*$PROD:Y(i)$(vom(i) and ele(i)) s:0  kle:sigma("e_kl") a(kle):sigma("k_l")
*+ ene(kle):sigma("fe") oil(ene):0 col(ene):0 gas(ene):0
*       O:PY(i)                            Q:vom(i)
*       I:PY(j)$(not fe(j))     Q:vfi(j,i)
*       I:PL                    Q:vfm("lab",i)                 va:
*       I:PK                    Q:vfm("cap",i)                 va:
*       I:PY("oil")             Q:vfi("oil",i)                 oil:
*       I:PCO2$clim             Q:fico2("oil",i)    P:1e-5     oil:
*       I:PY("col")             Q:vfi("col",i)                 col:
*       I:PCO2$clim             Q:fico2("col",i)    P:1e-5     col:
*       I:PY("gas")             Q:vfi("gas",i)                 gas:
*       I:PCO2$clim             Q:fico2("gas",i)    P:1e-5     gas:
```

A simple energy-economic model (ENERGY2.gms)

```
*       Electricity production
$PROD:Y(i)$(vom(i) and ele(i)) s:0  kle:sigma("e_kl") va(kle):sigma("k_l")
+ ene(kle):sigma("fe") oil(ene):0 col(ene):0 gas(ene):0
        O:PY(i)                          Q:vom(i)
        I:PY(j)$(not fe(j))      Q:vfi(j,i)
        I:PL                     Q:vfm("lab",i)                va:
        I:PK                     Q:vfm("cap",i)                va:
        I:PY(fe)                 Q:vfi(fe,i)                   fe.tl:
        I:PCO2#(fe)$clim         Q:fico2(fe,i)     P:1e-5      fe.tl:
```



Electricity

A simple energy-economic model (ENERGY2.gms)

```
*       Other output and refined oil production
$PROD:Y(i)$(vom(i) and not ele(i) and not ff(i)) s:0 kle(s):sigma("e_kl")
va(kle):sigma("k_l") e(kle):sigma("ele_fe") ne(e):sigma("fe")
+ oil(ne):0 col(ne):0 gas(ne):0
        O:PY(i)                         Q:vom(i)
        I:PY(j)$(not fe(j))     Q:vfi(j,i)                        e:$ele(j)
        I:PL                    Q:vfm("lab",i)                    va:
        I:PK                    Q:vfm("cap",i)                    va:
        I:PY(fe)                Q:vfi(fe,i)                       fe.tl:
        I:PCO2#(fe)$clim        Q:fico2(fe,i)     P:1e-5          fe.tl:
```



Other output

A simple energy-economic model (ENERGY2.gms)

```
*       Consumer welfare function
$PROD:W s:sigma("top_fd") e:sigma("ene_fd") oil(e):0 col(e):0 gas(e):0
        O:PW                   Q:vthp
        I:PY(i)$(not fe(i))    Q:vhp(i)                          e:$ele(i)
        I:PY(fe)               Q:vhp(fe)                         fe.tl:
        I:PCO2#(fe)$clim       Q:hhco2(fe)    P:1e-5             fe.tl:
```



Consumption

A simple energy-economic model (ENERGY2.gms)

```
*        Household demand
$DEMAND:HH
         D:PW                     Q:vthp
         E:PL                     Q:efh("lab")
         E:PK                     Q:efh("cap")
         E:PR(i)                  Q:vfm("res",i)
         E:PCO2$clim              Q:clim
```

- ENERGY2.gms implements a 20% reduction in CO2 emissions

```
clim = 0.8*sum(i, fco2(i));
$include ENERGY2.gen
solve ENERGY2 using mcp;
```

- All model output is sent to ENERGY2_OUTPUT.gdx
  - Useful for keeping track of parameter values and flags

A simple energy-economic model (`ENERGY2.gms`)

- Results reporting includes the quantity of emissions and the CO2 price
  - Economic data are in billions and emissions data are in millions of metric tons

```
PARAMETER
EMISSIONS(*)            CO2 emissions - million metric tons
PRICECO2(*)             CO2 price $ per metric ton;

EMISSIONS("BASE") = sum(i, fco2(i));

EMISSIONS("SIM") = sum((fe,i)$vfi(fe,i),
fico2(fe,i)*FFE.L(fe,i)/vfi(fe,i))
+ sum(fe, hhco2(fe)*HFE.L(fe)/vhp(fe));

EMISSIONS("%C") = 100*(EMISSIONS("SIM")-
EMISSIONS("BASE"))/EMISSIONS("BASE");

PRICECO2("SIM") = 1e3*PCO2.L/PW.L;
```

Exercises

1. Implement a 30% reduction in emissions and analyze the results.
2. Change the value of sigma("e_kl") from 0.5 to 1 and analyze changes in the CO_2 price and welfare.
3. Implement a sector-specific cap on electricity emissions (without restrictions on emissions from other sectors).

6. Adding an advanced technology to a computable general model

Advanced technologies

- Several low or zero-carbon energy technologies are currently available but are not profitable (at large scale) under current market conditions
  - Electricity from biomass
  - Electricity from wind
  - Electricity from solar
  - Electricity from fossil fuels with carbon capture and storage (CCS)
  - Refined oil from biofuels
- These technologies can be added to a CGE model is MPSGE format relatively easily
- As advanced technologies are typically not represented in the base date, characterizing production functions for these technologies require bottom-up estimates of production costs

Advanced technologies

- Bottom-up estimates typically include the cost of production and cost shares for capital, labor, intermediate inputs and an industry-specific fixed factor
- The fixed factor (and the elasticity of substitution between the fixed factor and other inputs) controls the penetration of the advanced technology
- The penetration of new technologies will be limited by factors such as:
  - The number of suitably-qualified engineers
  - The number of suitable sites for certain technologies, such as wind and solar
- It is convenient to specify input costs for advanced technologies so that they sum to one and apply a “markup” factor that represents the cost of production using the advanced technology relative to those for existing technologies

Adding an advanced technology in MPSGE

- `ENERGY3.gms` specifies a model that includes a representative renewable electricity production function
- Availability of the renewable technology is controlled using an "`ACTIVE`" flag

```
* Define and assign parameters for an advanced electricity generation technology
PARAMETER
  ACTIVE   Flag to make a renewable electricity technology available
  MARKUP   Cost of renewable electricity relative to conventional generation
  BSIN     Input cost shares for renewable electricity
  BRES     Renewable fixed factor resource;

ACTIVE = NO;
MARKUP = 1.2;
BSIN("othr") = 0.2;
BSIN("cap") = 0.5;
BSIN("lab") = 0.2;
BSIN("ffactor") = 0.1;
BRES = 0.01*vom("ele");
sigma("top_renew") = 0.2;
```

Adding an advanced technology in MPSGE

```
$SECTORS:
        W                       ! Welfare index
        Y(i)$vom(i)             ! Output index
        EB$ACTIVE               ! Renewable electricity index

$COMMODITIES:
        PW                      ! Welfare price index
        PY(i)$vom(i)            ! Output price index
        PL                      ! Labor price index
        PK                      ! Capital price index
        PR(i)$vfm("res",i)      ! Natural resources price index
        PCO2$clim               ! Price of co2 emissions
        PBF$ACTIVE              ! Price of fixed factor for renewable electricity

*       Household demand
$DEMAND:HH
        D:PW                    Q:vthp
        E:PL                    Q:efh("lab")
        E:PK                    Q:efh("cap")
        E:PR(i)                 Q:vfm("res",i)
        E:PCO2$clim             Q:clim
        E:PBF$ACTIVE            Q:BRES
```

Adding an advanced technology in MPSGE

```
*         Renewable electricity generation
$PROD:EB$ACTIVE s:sigma("top_renew") b:0 c(b):1
          O:PY("ele")    Q:1
          I:PY("oth")     Q:(BSIN("oth")*MARKUP) b:
          I:PL            Q:(BSIN("lab")*MARKUP)  c:
          I:PK            Q:(BSIN("cap")*MARKUP)  c:
          I:PBF           Q:(BSIN("ffactor")*MARKUP)
```



Adding an advanced technology in MPSGE

- The penetration of renewables can be calibrated to an estimated supply elasticity (ε) according to:

$$\varepsilon = \sigma_{F-O} \frac{(1-\alpha_F)}{\alpha_F}$$

Where α_F is the share of the fixed factor in total cost, and σ_{F-O} is the elasticity of substitution between the fixed factor and other inputs

- Under a cap-and-trade policy, the increase in the cost of generation from conventional (fossil fuel) sources allows renewable electricity generation to compete with conventional technologies

Adding an advanced technology in MPSGE

- Production of advanced technologies can also be induced by mandates (e.g., renewable fuel targets set out under RFS2)
- Mandates for renewable electricity can be modelled by assigning a permit for each unit (MWh) of renewable electricity produced and requiring utilities to turn in a certain number of allowances (α) for each unit of electricity sold





Exercises

1. Change the markup from 1.2 to 2 and explain the outcome you observe.
2. Using a markup of 1.2, change sigma("top_renew") from 0.5 to 0.2 and compare changes in renewable electricity production and the CO_2 price.
3. Without a cap-and-trade policy, implement a renewable portfolio standard so that 5% of total electricity generation is from renewable generation. Set the markup equal to 1.2 and sigma("top_renew") equal to 0.5.

7. A global model for climate policy analysis

Why use a global model?

- As most nations are interconnected, a global model (or at least a model that considers international trade flows) is essential for understanding the impact of climate policies
- Global models facilitate assessment of how policies in one region influence policies in other regions
  1. Terms-of-trade changes
  2. Leakage of emissions (increased emissions in unconstrained regions due to climate policies in constrained regions)

The Global Trade Analysis Project

- The Global Trade Analysis Project (GTAP) database was established in 1993 by the Department of Agricultural Economics at Purdue University
  - "Contains detailed bilateral trade, transport and protection data characterizing economic linkages among regions, together with individual country input-output databases which account for intersectoral linkages within regions"
- The most recent version (version 8) of the database corresponds to the world economy in 2007
- Identifies 129 regions, 57 sectors and 7 factors of production
- 14 agricultural sectors, 4 resource-based sectors, 8 food sectors, 16 manufacturing sectors, and 15 service sectors
- Land, coal resources, oil resources, gas resources, capital and skilled and unskilled labour
- More information is available at: https://www.gtap.agecon.purdue.edu/

A global CGE model

- The “standard” GTAP model is a static, multiregional model of the global economy that determines the production and allocation of goods
- Sectors are perfectly competitive
- Interactions between regions are captured by a full set of bilateral trade flows
- Two-way trade is modeled using the “Armington assumption”
  - Each region produces a unique variety of each good
  - Consumption of each good is a nested CES aggregate of imported and domestically produced varieties
- A representative consumer maximises utility by allocating expenditure across private consumption, government consumption, and investment

A global CGE model



A global CGE model

- The file `GLOBAL.gms` implements a model built on a 2x7 aggregation of the database
- Regions: The climate coalition (OECD countries) and the non-coalition (non-OECD countries)
- Sectors: coal, gas, crude oil, refined oil, electricity, energy-intensive industry (EINT), other industry (OTHR)

A global CGE model

- Transportation services for commodities traded internationally are a CES aggregation of transportation services from different regions

```
$prod:yt(j)$vtw(j)  s:1
        o:pt(j)         q:vtw(j)
        i:py(j,r)       q:vst(j,r)
```



A global CGE model

- Imports to each region are a CES aggregation of imports from different regions (including transport costs)

```
$prod:m(i,r)$vim(i,r)   s:esubm(i)  s.tl:0
        o:pm(i,r)       q:vim(i,r)
        i:py(i,s)       q:vxmd(i,s,r)   p:pvxmd(i,s,r) s.tl:
+               a:hh(s) t:(-rtxs(i,s,r))
+               a:hh(r) t:(rtms(i,s,r)*(1-rtxs(i,s,r)))
+               n:tau(i,s,r)$border(i,s,r)
        i:pt(j)#(s)     q:vtwr(j,i,s,r) p:pvtwr(i,s,r) s.tl:
+               a:hh(r) t:rtms(i,s,r)
```

- In each region, each commodity entering final demand is a CES aggregate of a (composite) imported variety and a domestic variety

```
$prod:A(i,r)$a0(i,r)    s:esubd(i)
        o:pa(i,r)       q:a0(i,r)
        i:py(i,r)       q:d0(i,r)      p:(1+t_d(i,r)) a:hh(r) t:t_d(i,r)
        i:pm(i,r)       q:m0(i,r)      p:(1+t_m(i,r)) a:hh(r) t:t_m(i,r)
```

A global CGE model



A global CGE model

- `GLOBAL.gms` considers (i) a cap that reduces coalition emissions by 20%, and (ii) border carbon adjustments (tariffs on embodied CO_2 emissions) on imports from the coalition
- Border carbon adjustments have been proposed to address leakage and competitiveness concerns
- Leakage of emissions occurs via at least two channels
  - Trade: Substitution towards goods and services produced in unconstrained regions
  - Fossil fuel prices: Reduced demand for fossil fuels in constrained regions decreases the prices of these commodities and increases fossil fuel use in unconstrained regions

(i) Reducing emissions by 20%

```
clim(anx) = 0.8*ghgt("BAU",anx);
```

The leakage rate is defined as:

$$100\,\frac{(Increase\ in\ emissoins\ in\ unconstrained\ regions)}{(Decrease\ in\ emissoins\ in\ carbon\ constrained\ regions)}$$

```
leakage("CAT") = 100*(ghgt("CAT","nclt") - ghgt("BAU","nclt")) /
                 (ghgt("BAU","clt") - ghgt("CAT","clt"));
```

(ii) Imposing border carbon adjustments

```
$auxiliary:
        tau(i,s,r)$border(i,s,r) ! Border taxes on carbon content



$prod:m(i,r)$vim(i,r)   s:esubm(i)  s.tl:0
        o:pm(i,r)       q:vim(i,r)
        i:py(i,s)       q:vxmd(i,s,r)   p:pvxmd(i,s,r) s.tl:
+               a:hh(s) t:(-rtxs(i,s,r))
+               a:hh(r) t:(rtms(i,s,r)*(1-rtxs(i,s,r)))
+               n:tau(i,s,r)$border(i,s,r)
        i:pt(j)#(s)     q:vtwr(j,i,s,r) p:pvtwr(i,s,r) s.tl:
+               a:hh(r) t:rtms(i,s,r)


 $constraint:tau(i,s,r)$border(i,s,r)
    tau(i,s,r)*py(i,s) =e= pcarbon(r)*eco2(i,s);
```

(ii) Imposing border carbon adjustments

```
$auxiliary:
        tau(i,s,r)$border(i,s,r) ! Border taxes on carbon content



$prod:m(i,r)$vim(i,r)   s:esubm(i)  s.tl:0
        o:pm(i,r)        q:vim(i,r)
        i:py(i,s)        q:vxmd(i,s,r)   p:pvxmd(i,s,r) s.tl:
+               a:hh(s) t:(-rtxs(i,s,r))
+               a:hh(r) t:(rtms(i,s,r)*(1-rtxs(i,s,r)))
+               n:tau(i,s,r)$border(i,s,r)
        i:pt(j)#(s)      q:vtwr(j,i,s,r) p:pvtwr(i,s,r) s.tl:
+               a:hh(r) t:rtms(i,s,r)


 $constraint:tau(i,s,r)$border(i,s,r)
    tau(i,s,r)*py(i,s) =e= pcarbon(r)*eco2(i,s);
```

(ii) Imposing border carbon adjustments

- For each sector, embodied emissions are calculated as the sum of emissions associated with direct fossil fuel use plus emissions from electricity used by that sector

```
*       Calculate embodied CO2 emissions
Parameter
elec_s(i,r)      Electriciyt emissoins associted with each sector
eco2(i,r)        Embodied CO2 emissions - direct plus electricity emissions;

elec_s(i,r) = sum(fe, eghg("co2",fe,"ele",r))*vaf("ele",i,r)/vom("ele",r);

eco2(i,r)$vom(i,r) = (sum(fe, eghg("co2",fe,i,r)) + elec_s(i,r))/vom(i,r);
```

Exercises

1. Double the elasticities of substitution in the Armington nest and compare the results to the base case.
2. Calculate indirect CO_2 emissions as the sum of emissions embodied in all intermediate inputs (not just electricity) and compare your results to the base case.

Important characteristics of EPPA not covered

1. The Dynamic structure: How the model evolves over time
   - Population, labor productivity, the capital stock and resource depletion change over time
   - Activity in one period affects the availability of capital and resources in future periods
2. Vintaging of the capital stock
   - Only a proportion of the capital stock is mobile across sectors and some capital is “frozen” into existing production techniques